
Abstract


This presentation might be considered introductory, but covers an area of Vision system configuration that is commonly misunderstood: the use and understanding of logic to effect delays and detect transition events.  Guidelines and examples are presented which should be of benefit to users of all experience levels.








Introduction


In the typical process control environment a major portion of the system processing is based on logic which is repetitive or rate based.  At periodic time intervals, logic and calculations are executed, with the evaluations being based on current or "real-time" statuses.  An example is the alarm condition test such as "is the current reading above/below the alarm limit".





There are other circumstances in a process control system environment which require data or status's from the previous or prior history status.  Examples include simple "change of state" events, such as push button scheduling of an event based transmission to a Unitec processor; running of logic/algorithms only on a certain transition; delaying a transitional state, such as might be used for debounce.  Such event or delay processing can be implemented at either the Unitec or VPU level.








Unitec Environment


DOP - transition statements


The DOP processor contains various statements which accomplish delay or transition processing logic.  In this context, transition means change of state or difference between the current status (or value) and the previous status.  To briefly review these, there are three basic transition processing logic statements:





ON LSH1234cio TR01 THEN  ! This evaluates true(executes) when


...                      ! CI/CO/FLAG transitions from OFF to ON


�
�
ON LSH1234cio TR10 THEN  ! This evaluates true when an 


...                      ! ON to OFF transition occurs


�
�
ON LSH1234cio TR   THEN  ! This evaluates true for either 


                         ! transition occurrence.


�
�
(where LSH1234cio could be a contact in/out/flag on a DOP)�
�



Notice that this refers to contact inputs, outputs or flags only.  These are true Boolean (i.e.. ON/OFF) devices on the DOP board.  DOP logical devices (i.e., valves, pumps, timers and counters), represent their status (internally at least) as an integer, from 1 through 17.  Despite any perception we might have of the status of a pump or a valve being a Boolean state, DOP logical devices have one of several possible states.   Using an integer allows several possible states to be represented, including on, off, open, closed, start, run, done, reset, etc.  This means that for logical devices, the only legal transition code that will work is:





ON TMR1234lio\SENSE TR THEN  ! This evaluates true on any status


                             ! transition of the timer device.


�
�
(where LSH1234cio could be a contact in/out/flag on a DOP)�
�



An important point here is the inclusion of the qualifier "\SENSE" following the device name (IO master id) of the timer device.  The sense state of the device is the only state transition that is detected by this logic construct.  IF the user erroneously uses the "\COMMAND" qualifier (or leaves the spot blank) the compiler will still detect the sense state.





For those unfamiliar with logical devices, they have both a "command" status and a "sense" status.  The command status contains the last operator or logic issued command to the device ON; OFF; OPEN; CLOSE; START; STOP.  The sense state has the additional capacity to contain either TRANSITION or FAILURE statuses, based on the feedback status and stroke time elapsed.  If the feedback contact input(s) doesn't confirm or agree with the command state before the stroke timer elapses, the sense state becomes FAILURE .  TRANSITION status occurs on valves with two feedback contact inputs, when neither input indicates a confirmation.





Of course, that doesn't address how to detect specific transitions of logical devices.  Illustrated below is a timed charge using DOP logic.  The logic makes use of a DOP timer device.  The flag CHRG1201.ST is the start flag and its transition from off to on initiates the charge by opening the valve and resetting (zeroing) then starting the timer device.   





�
DOP Program device code:


! DOP BOARD D310717PIO PROGRAM DEVICE


!             Flag declarations


CHRG1234.ST   D310717F1FLG\FLAG_1; ! TK1201- start timed charge





!--------------------------------------------


!             TRANSITION LOGIC


!--------------------------------------------





ON CHRG1234.ST TR01 THEN           ! Start Trans ON, start charge


   FCV1234lio\COMMAND = OPEN;      ! and open charge valve


   TMR1234lio\COMMAND = RESET;     ! reset timer to 0


   TMR1234lio\COMMAND = START;     ! start timer counting


ENDIF;





The next sequence of logic is written to detect any transition on timer TMR1234LIO, but then tests for state being DONE or STOPed.  This effectively allows the detection of any logical device state transition.





DOP Program device code (cont):





ON TMR1234lio\SENSE TR THEN         ! timer transition, test which


   IF TMR1234lio\SENSE EQ DONE OR   ! mode just occurred


      TMR1234lio\SENSE EQ STOP THEN ! if done or stop,


      CLEAR CHRG1234.ST;            ! clear charge flag


   ENDIF;                           ! which will close valve


ENDIF;





ON CHRG1234.ST TR10 THEN            ! Start Trans OFF, Stop charge


   TMR1234lio\COMMAND = STOP;       ! stop timer count


   FCV1234lio\COMMAND = CLOSE;      ! and close charge valve


ENDIF;











VPU Environment


Button Scheduled Blocs


We see that on a DOP board, the usage of event transition refers to detecting the transition of the on/off status of contact inputs, outputs or flags, and can also be used to detect logical device state transitions.  Since we are on the subject of transitions and this relates to our example, let's examine another type of event transition, the push button.  This illustrates the button code and supporting algorithm to effectively send the start or stop command directly to the flag CHRG1234.ST, shown above.  The first illustration is the button logfile code, which shows button 3 setting the ordinal ...STR while button 13 sets the ordinal ...STP.  Both buttons schedule the same button event.  The schedule command will cause the chain(s) linked to that event to be run.





�
Button Code


! Button logfile for shot timer start & stop


!


CRE,FIF,


SET,1234,NONE,NO,


TITLE,START & STOP SHOT TIMER 1234





select,3,


label,60 SECOND,SHOT TIME,


ord,set,CHRG1234.STR,


! schedule,fex,fex,EV_CHRG_BTN,EXIT,


schedule,fex,fex,(1,,0,,207,,EVT),EXIT,





select,13,


label,SHUTDOWN,SHOT TIME,


ord,set,CHRG1234.STP,


! schedule,fex,fex,EV_CHRG_BTN,EXIT,


schedule,fex,fex,(1,,0,,207,,EVT),EXIT,





EXIT,NO,


(Special considerations must be given if scheduling events across the Data Freeway to other VPU's.  When specifying an event, Button builder does not allow symbolic entry of LPN id prefixing the event name.  Instead the event must be specified using the format (1,0,207,evt), where 1 is LPN1 and 207 is the listed numeric contents of the event symbol from BLOC builder.  This number is determined by listing the event in block builder which will reveal the event number.  For example:


BLOC:   LIST EVENT EV_CHRG_BTN 


     NAME:  ev_chrg_btn


     LPN:  lpn1


     BCT:  bc_chrg_btn


     NUMBER:  207


     TYPE:  (SCHED,MEMRES)


     PRIORITY:  100





 





The BLOC builder code to create and support the event referred to from the button is illustrated below.  In actual practice the events and BCT's must be entered first since the button must refer to an already existing event.  The BCT is entered first then the event is declared.  These two go together and will always have a one to one relationship (you should never find more than one event pointing to a specific BCT).





BLOC: Code 


SET DEF LPN LPN1


ADD BCT   BC_CHRG_BTN,CHAINRES,"BCT for button sched event"


ADD EVENT SCHED,MEMRES,EV_CHRG_BTN,BC_CHRG_BTN,100,"event for button sched"








In this example, one chain is created for dynatouch button event scheduling within the logical area related to vessel charging, named "CH_CHRG_BTN".  Smaller systems might only need to use one event per LPN.  





The general purpose block illustrated below is solely to support the PTSREDMSG algorithm which sends an acknowledge response to the Video Control Task, which then releases the dynatouch button lockout/debounce delay.





ADD CHA CH_CHRG_BTN,,BC_CHRG_BTN,btib1,1,1,1,(COLD,TEPID),btid1,btid1\$l.on





! Block for purpose of sending ptsredmsg ACK back to VCT


ADD BLO CH_CHRG_BTN\BL_CHRG_BTN,,1,btid1\$l.on,ACTIVE


ADD ALG CH_CHRG_BTN\BL_CHRG_BTN,,ptsredmsg,4,0








The application blocks perform two tasks, the first sends a set ($L.ON) to a flag to start the timed output, while the second sends a clear ($L.OFF) command, used to abort or shutdown the timed charge before the set time had finished.





SET DEF LPN LPN1


ADD SYMBOL CHRG1234_OUT,600.0,(VAR,REAL),BTID1,"value for 60 secs"


ADD SYMBOL CHRG1234.STR,OFF,ORDINAL,BTID1,"ordinal for start shot"


ADD SYMBOL CHRG1234.STP,OFF,ORDINAL,BTID1





! Block to initiate the timed shot, output to TMR1234lio & set CHRG1201.ST


ADD BLO CHN_CHRG_BTN\BL_CHRG1234,,1,CHRG1234.STR,ACTIVE





ADD ALG  CHN_CHRG_BTN\BL_CHRG1234,,ONESHOT





ADD ALG  CHN_CHRG_BTN\BL_CHRG1234,ONESHOT,OUTPUT,D310717.BD,-


"DIR_IOM",D310717F1FLG,1,"ORDINAL",BTID1\$L.ON,D310717F1S01





! Block to shutdown the timed shot, clear flag CHRG1201.ST.


ADD BLOCK CHN_CHRG_BTN\BL$CHRG1234,,1,CHRG1234.STP,ACTIVE





ADD ALG  CHN_CHRG_BTN\BL$CHRG1234,,ONESHOT





ADD ALG  CHN_CHRG_BTN\BL$CHRG1234,ONESHOT,OUTPUT,D310717.BD,-


"DIR_IOM",D310717F1FLG,1,"ORDINAL",BTID1\$L.OFF,D310717F1S01








Unitec Environment


PIDP/ICP/DOP - coded transition detection


There are other situations where you might need to act upon a transition detected from within rate based logic, where DOP transition detection statements are not available or appropriate.  This could occur on any Unitec program device, since the PIDP and ICP have no built in transition statements, or on the DOP if the code has a nested logic requirement.  Process specification such as:


if hi level alarm transitions on 


      stop pump, 


      then 5 seconds after pump transitions off


            close valve


would require transition detection within rate based (scheduled) logic.  





To detect a transition without benefit of a transition statement, you simply create an extra flag (DOP board) or Boolean variable (PID/ICP boards) for a history flag and code the statement to detect the desired difference or transition.  After the comparison, you must copy the current value into the history flag so each subsequent execution will evaluate the change between the current time and the last execution.  





Also, consideration should be given to the appropriate initialization logic for Unitec reset condition.  After a program device reset, the usual action is to clear any status or history flags, since resetting the program device probably disrupted any logic in progress.  





For the sake of comparison, similar logic is illustrated as it would be implemented on either the DOP or PIDP program devices.





DOP Program device code:


                                     ! Flag declarations


CHRG1201.ST   DOP101F01FIO\FLAG_1;   ! TK1201- start timed charge


CHRG1201.Hi   DOP101F01FIO\FLAG_2;   ! TK1201- history for start charge





!--------------------------------------------


!             Initialization Logic


!--------------------------------------------


ON INIT THEN


   CLEAR CHRG1201.ST;                ! After reset, variable state is 


   CLEAR CHRG1201.Hi;                ! unknown and must be cleared.


ENDIF;


!--------------------------------------------


              SCHEDULE               ! Start of Scheduled logic


!--------------------------------------------





IF CHRG1201.ST XOR CHRG1201.Hi THEN  ! This detects either transition


   FCV1299lio\COMMAND = CLOSE;       ! close drain valve


ENDIF;


CHRG1201.Hi = CHRG1201.ST;           ! keeps history tracking current


�
For those unacquainted with it, the "XOR" comparison is an "exclusive or" which evaluates true if the two logical (or Boolean) variables are different.  The statement could be translated:


"IF CHRG1201.ST is different from CHRG1201.Hi THEN"





The detection would therefore be either the off to on transition or on to off.  The more typical use is the detection of one direction.  The two possibilities are shown below:





                                        ! This detects a 0 to 1 transition


IF CHRG1201.ST AND NOT CHRG1201.Hi THEN ! i.e. an OFF to ON transition


   FCV1234lio\COMMAND = OPEN;           ! then open charge valve


   TMR1234lio\COMMAND = START;          ! and start timer counting


ENDIF;


CHRG1201.Hi = CHRG1201.ST;              ! keeps history tracking current





                                        ! This detects a1to 0 transition


IF NOT CHRG1201.ST AND CHRG1201.Hi THEN ! i.e. an ON to OFF transition


   FCV1234lio\COMMAND = CLOSE;          ! then close charge valve


   TMR1234lio\COMMAND = STOP;           ! and timer counting


ENDIF;


CHRG1201.Hi = CHRG1201.ST;              ! keeps history tracking current








For a comparison, on the PIDP or the ICP program devices, the same logic would be implemented:


PIDP Program device code:


LOCAL_BOOLEAN                        ! Variable declarations


   CHRG1201.ST, ! TK1201- start timed charge (from VPU Button push)


   CHRG1201.Hi; ! TK1201- history for start charge





!--------------------------------------------


!             Initialization Logic


!--------------------------------------------


IF NOT VALID(CHRG1201.Hi) then       ! After reset, variable is invalid


   CLEAR CHRG1201.ST;                ! and must be cleared


   CLEAR CHRG1201.Hi;


ENDIF;





!--------------------------------------------


!             SCHEDULED CODE


!--------------------------------------------


IF CHRG1201.ST XOR CHRG1201.Hi THEN  ! This detects either transition


   CLEAR FIC1201DIO\DDCREQ;          ! Put loop into manual mode


   IF NOT FIC1201DIO\DDCON THEN      ! test if it got there ?


      FIC1201DIO\OUTTARGO = 2.0;     ! once it does, you can set the PO.


      CHRG1201.Hi = CHRG1201.ST;     ! keeps history tracking current 


   ENDIF;                            ! only once the loop is in manual


ENDIF;                               ! since that stops the loop.








As with the DOP example, this detects either transition, but we are doing something more interesting when it occurs.  The logic puts the controller in manual, then when confirmed by feedback that it's in manual, set the PO = 0%.  This insures that the controller will accept the PO entry and not conflict with the still active PID control.





To finish the PID/ICP transition possibilities, the off to on and the on to off transition are shown below:


                                        ! This detects a 0 to 1 transition


IF CHRG1201.ST AND NOT CHRG1201.Hi THEN ! i.e. an OFF to ON transition


   CLEAR FIC1201DIO\DDCREQ;             ! Put loop into manual mode


ENDIF;


CHRG1201.Hi = CHRG1201.ST;              ! keeps history tracking current 





                                        ! This detects a 1 to 0 transition


IF NOT CHRG1201.ST AND CHRG1201.Hi THEN ! i.e. an ON to OFF transition


   CLEAR FIC1201DIO\DDCREQ;             ! Put loop into manual mode


ENDIF;


CHRG1201.Hi = CHRG1201.ST;              ! keeps history tracking current 


�



VPU Environment


Ordinal Transition Detection


The preceding illustrations showed rate based logic being used for transition or difference detection at the Unitec level.  The same concepts would be applicable at the BLOC level using MPL algorithms.  Additionally, BLOC contains two other useful methods of detecting an ordinal transition directly.  The simplest is the algorithm "ONTRANS", which prompts the user for ordinal and type of transition.  If the algorithm detects the specified transition, the algorithm passes control to (executes) the remaining algorithms in the block.  If no transition is detected, the remaining algorithms in the block are skipped.  





SET DEF LPN LPN1


ADD SYMBOL CHRG1234_FB,OFF,ORDINAL,LPN1\BTID1,"ord for shot feedback"


                                         ! Detect  OFF to ON transitions


ADD ALG CHN_3_SEC\BLK_CHRG1220,,ORDTRANS,CHRG1201_FB,"TR01"





Alternatively, you may specify "TR10" for ON to OFF transition 


or "TR0110" for either transition detection.





This would be typically used in relatively fast (3 seconds in this example) rate based logic so that a difference would be detected fairly soon after the actual transition.  The other approach is the STEPFN algorithm, which has the same transition detection options in its comparison logic.








Unitec Environment


DOP - Delay Statements Logic


One other major consideration for advance logic is the use of the different forms of delay statements.  At the Unitec level, the DOP board is the only board currently able to directly support delay logic.  This takes the form of three possible statements DELAY(); TDELAY(); and FDELAY().  Delays can also be effected by using DOP resident timer logical devices, which we have seen in the code earlier.  For PIDPs or ICPs the only option to implement a delay is to code a counting loop and a limit test.  Indirectly an ICP can access and have control of any DOP board timer devices.  





An aggravation with the otherwise very useful timer devices is that the device's reported elapsed time count is a 32 bit long integer, in 100 msec units.  Unless you have operators who are conformable with entering 1200 for 2 minutes, you will need to provide math (MPL) to convert their entries from minutes to (seconds * 10) before transmitting to the device.  Likewise, the reported value is not easily converted to minutes, you will either convert on the ICP (if needed at the Unitec level), or alternatively use the "CONVERT" algorithm, to convert the long integer to a real number.





The three DOP delay statements are illustrated below.  The delay statements (DELAY/TDELAY/FDELAY)  allow the user to specify a contact input/output or flag as well as a delay count.  The on/off state is put into a history stack to a depth specified by the delay count.  Special consideration is given to logical devices, due to its state being an integer.  A logical device state can be delayed by one count only, using the pure DELAY statement.  Since logical devices are not true or false (i.e. Boolean), they are not supported by the TDELAY or FDELAY statements.





It is important to note that a specified delay is not directly related to time, but is instead execution cycles, so a delay of 9 means 9 execution cycles of the statement.  If the delay statement is nested within a conditional clause which is true only once an hour, the count of 9 will be 9 hours.  





These delay statements can only be used, and only make sense when used, in the scheduled logic area on the DOP.  In other words, you cannot successfully use a delay within an "ON ... TR THEN" clause since delay statements update delay counters on each program execution.  (The compiler would prevent this as well).  If the delay of a transition occurrence is needed, this can be accomplished using a combination of the described logic.





Let's consider the delay statements in some detail.  First the format of a delay is DELAY/TDELAY/FDELAY(VARIABLE, COUNT); , where VARIABLE is a contact input/output or flag and COUNT is an integer.  As mentioned in the specific case of DELAY, VARIABLE could also be a logical device.





DELAY


The pure DELAY statement is a simple postponement of the state of the specified input.  Since the delay count is in execution cycles, the code 


"XYZ = DELAY(LSL1234CIO,8);" could be stated as "XYZ contains the state of LSL1234CIO  8 cycles ago".  The state of the input value compared to the delayed output is illustrated.





�


Delay statement (DELAY) timing


�
TDELAY


By comparison, using a true delay will delay only the true transition of the input, while the false (or off) condition is output immediately.  The code "XYZ = TDELAY(LSL1234CIO,8);" could be stated "XYZ is true if LSL1234CIO is true and has been for 8 cycles (continuously), otherwise XYZ is false".





�


True Delay (TDELAY) statement timing








FDELAY


Conversely, a false delay will delay only the false transition of the input, while the true state is output immediately.  This is not the inverse of a true delay.  The code "XYZ = FDELAY(LSL1234CIO,8);" could be stated "XYZ is FALSE if LSL1234CIO is false and has been for 8 cycles, otherwise XYZ is true".  This means that "FDELAY(abc)" is quite different than "NOT TDELAY(abc);".








�


False Delay (FDELAY) statement timing





�
It is worth illustrating the NOT TDELAY graph, since the difference between a "NOT TDELAY" and an "FDELAY" is frequently misunderstood.  Notice the difference between the resulting output signal from the "NOT TDELAY"(below) compared to the "FDELAY" (bottom of previous page).





�


Not True Delay (NOT TDELAY) statement timing





Examples


By combining delay timing, you can create a debounce function, which will not respond to short duration pulses or chatter, but with a capability of much greater than the 2.5 seconds provided by the DOP board firmware.  





In this example, a sump pump control is triggered by a high level switch, LSH1234, which must be on continuously for 60 program execution cycles (approximately 15 sec ) before setting the pump run flag.  After the high level switch clears and stays off for 40 cycles continuously, the statement "NOT FDELAY (LSH1234CIO,60)" evaluates true and the pump run flag is cleared.





�


Combined Delay timing





�
Let's consider some examples showing the use of these delay statements.  One example is an improvement on code used earlier, to detect a transition, with the history flag.  This is still the method of choice for PIDP or ICP boards, but on a DOP we can eliminate the history flag and use a delay statement instead.  This avoids consuming DOP flags.





DOP Program device code:


                                        ! Flag declarations


CHRG1201.ST   DOP101F01FIO\FLAG_1; ! TK1201- start timed charge





!--------------------------------------------


              SCHEDULE


!--------------------------------------------


                                        ! this detects either transition


IF CHRG1201.ST XOR DELAY( CHRG1201.ST,1 ) THEN 


   FCV1234lio\COMMAND = OPEN;           ! then open charge valve


   TMR1234lio\COMMAND = START;          ! and start timer counting


ENDIF;





                            ! This detects a 0 to 1 transition (OFF to ON )


IF CHRG1201.ST AND NOT DELAY( CHRG1201.ST,1 ) THEN 





                            ! This detects a1to 0 transition (ON to OFF )


IF NOT CHRG1201.ST AND DELAY( CHRG1201.ST,1 ) THEN 








It was stated that delays cannot be placed in an ON... TR THEN block, however the use of the delay statement can delay the transition by any reasonable multiple of the execution cycle.  In this example the requirement is to delay the transition of a flag by a count of 12 and perform the logic once, 12 cycles after the transition.





DOP Program device code:


!             Flag declarations


CHRG1201.ST   DOP101F01FIO\FLAG_1; ! TK1201- start timed charge





!--------------------------------------------


              SCHEDULE


!--------------------------------------------


                 ! This detects a previous transition (12 cycles ago) either direction


IF DELAY( CHRG1201.ST,12 ) XOR DELAY( CHRG1201.ST,13 ) THEN


   FCV1234lio\COMMAND = OPEN;      ! opens charge valve


   TMR1234lio\COMMAND = START;     ! start timer counting


ENDIF;





                 ! While this detects a 0 to 1 (i.e. an off to on) transition


IF DELAY( CHRG1201.ST,12 ) AND NOT DELAY( CHRG1201.ST,13 ) THEN





                 ! and this detects a 1 to 0 (i.e. an on to off) transition


IF NOT DELAY( CHRG1201.ST,12 ) AND DELAY( CHRG1201.ST,13 ) THEN 








Considering the example of the sump pump mentioned above, the state of the high level switch (named "LSH1234CIO") is the controlling influence.  LSH1234CIO is delayed in the scheduled logic, which is positioned in the later part of the DOP program device.  As the comments indicate, the high level switch must be true 15 seconds continuously to start the pump, and off the same amount before the pump is stopped.  This provides a hysteresis or debounce effect to insure that the pump will not start until the hi level is true for 15 seconds continuously, and also that once running, it will not switch off until the hi level is clear for 15 seconds.





DOP Program device code:


! Flag declarations


PU1201.ST     DOP101F01FIO\FLAG_1;





!--------------------------------------------


                  SCHEDULE


!--------------------------------------------


!********************************************


!*  HIGH Level SW Delay logic


!********************************************


IF TDELAY(LSH1234CIO,60) THEN      !-----------------------------


   SET PUMP1234.STR;               ! IF LSH1234 ON (= HI) for 15


ENDIF;                             ! sec continuously, start pump


                                   !----------------------------


IF NOT FDELAY(LSH1234CIO,60) THEN  ! IF LSH1234 OFF for 15


   CLEAR PUMP1234.STR;             ! sec continuously, stop pump


ENDIF;                             !----------------------------








The flag "PUMP1234.STR" is a start request flag.  In this example, the logic starts or stops the pump only once per flag transition and permits the operator to command the pump at any other time.





DOP Program device code:


!--------------------------------------------


!             TRANSITION LOGIC


!--------------------------------------------


!********************************************


!*  Low Level Sw LSH1234 Pump control logic *


!********************************************


                             !-----------------------------


ON PUMP1234.STR TR01 THEN    ! When the pump start flag 


   PUMP1234LIO\COMMAND = ON; ! transitions from off to on,


ENDIF;                       ! start pump PU1201


                             !-----------------------------


ON PUMP1234.STR TR10 THEN    ! When the pump start flag


   PUMP1234LIO\COMMAND = OFF;! transitions from on to off,


ENDIF;                       ! stop pump PU1201


                             !-----------------------------





�
Unitec Environment


DOP - Delay Statements Memory usage


The DOP delay statements do have a major drawback.  The history tables that these statements use can take a significant amount of the DOP board program device memory.  The only way to decide on limits is to know what the consumption is.  To that end, the following chart is provided.


The DOP program device has been configured with a delay statement.  The byte usage for various delay counts follow:  





DEVICE CONFIGURATION DATA -





SCHEDULE


LSH1234.DLY = DELAY(ZS5763D3CIO,8); ! DELAY 8 COUNT


END;





CONFIGURATION MEMORY:


USED        7   BYTES





A comparison of delay counts viruses byte usage.


DELAY COUNT�
BYTES USED�
�
15�
 7�
�
16�
12�
�
32�
17�
�
64�
27�
�



This works out to 5 bytes for each additional delay count of 16, or the formula of BYTES USED = (0.31 * DELAY COUNT).  Don't be intimidated about using these delays, just bear this consumption in mind.  





If your delay needs exceed several minutes and are for TDELAY or FDELAY statements, you may want to use a nested delay strategy.  In the following example, an FDELAY is used to set a flag once every 240 cycles (approx. 1 minute), then using that flag, a conditional run another FDELAY which by using a delay count of 59, sets another flag every 60 minutes.  Flush logic runs to toggle a valve every hour.





DOP37.1MIN    DOP37FLG1FIO\FLAG_1;


DOP37.60MIN   DOP37FLG1FIO\FLAG_2;


!--------------------------------------------------------------


!    Logic to create a 1 minute flag which will be on for .25 sec 


!    then off for 59.75 seconds. For timing delays 


!--------------------------------------------------------------


IF NOT FDELAY(DOP37.1MIN,239) THEN


   SET DOP37.1MIN;


ELSE


   CLEAR DOP37.1MIN;


ENDIF;


�
!--------------------------------------------------------------


! The following runs once a minute (based on DOP37.1MIN being true)


!--------------------------------------------------------------


IF DOP37.1MIN THEN


   IF NOT FDELAY(DOP37.60MIN,59) THEN


      SET DOP37.60MIN;


   ELSE


      CLEAR DOP37.60MIN;


   ENDIF;


ENDIF;


!--------------------------------------------------------------


! The following logic runs once an hour (when DOP37.60MIN is true)


!--------------------------------------------------------------


IF DOP37.60MIN THEN


   IF NOT FV1234.FLUSH THEN


      FV1234LIO\COMMAND = OPEN;


      SET   FV1234.FLUSH;


   ELSE


      FV1234LIO\COMMAND = CLOSE;


      CLEAR FV1234.FLUSH;


   ENDIF;


ENDIF;


!--------------------------------------------------------------


                   ! Code must clear 60 min flag. Otherwise it will be


CLEAR DOP37.60MIN; ! on for 1 full minute. We want it off after 1/4 sec 











Summary


We have explored several aspects of Vision system programming environment, as it related to transition detection and delay statement processing.  

